ME201/MTH281/ME400/CHE400
Coffee Cup Oscillations

1. Introduction

In this notebook, we study the modes of coffee oscillating in a coffee cup. These motions are an instance of what are called water waves in fluid dynamics. Such waves are likely to be seen whenever we have a liquid with a free surface in the presence of gravity. At equilibrium the free surface of the liquid coincides with a gravitational equipotential surface. When the surface is perturbed, an oscillation is set up in which the energy cycles between kinetic energy and gravitational potential energy. If the amplitude of the surface oscillations is small (the only case we consider here), the motions may be described by a linear theory. The most unusual aspect of this theory is that the governing equation is not a wave equation, but rather the Laplace equation, which normally serves as a model for equilibrium and not wave motion. The basic quantity in the theory is the velocity potential Φ, defined so that the fluid velocity $\mathbf{V}$ at any point is given by

$$\mathbf{V} = \nabla \Phi . \tag{1}$$

The existence of such a potential is a point developed at length in basic courses in fluid dynamics. We take it as a given here.

2. Governing Equations

In the present case, we consider water waves in a coffee cup -- a cylindrical container of circular cross section, with symmetry axis parallel to gravity. Let the radius of the cup be a, and let h be the equilibrium height of the coffee in the cup. Then the governing Laplace equation in cylindrical coordinates (r, θ, z) is

$$\frac{1}{r}\frac{\partial}{\partial r}\left(r\frac{\partial \Phi}{\partial r}\right) + \frac{1}{r^2}\frac{\partial^2 \Phi}{\partial \theta^2} + \frac{\partial^2 \Phi}{\partial z^2} = 0,\ 0 < r < a,\ 0 \leq \theta \leq 2\pi,\ \text{and}\ 0 < z < h\ , \tag{2}$$

where $z = 0$ is the rigid bottom of the cup, $z = h$ is the upper free surface of the coffee, and $r = a$ is the rigid side of the cup. The boundary conditions on the bottom and side of the cup come from the physical fact that the fluid cannot flow into a rigid surface. Thus the normal component of velocity must vanish on those surfaces. Hence

$$\frac{\partial \Phi}{\partial z}(r, \theta, 0, t) = 0,\ \text{and}\ \frac{\partial \Phi}{\partial r}(a, \theta, z, t) = 0. \tag{3}$$

The condition on the upper free surface is somewhat more involved, and it comes from both a kinematical analysis of the surface motion, and the time-dependent version of the famous Bernoulli equation. After some analysis, the condition may be put into the following form:

$$\frac{\partial^2 \Phi}{\partial t^2}(r, \theta, h, t) + g\frac{\partial \Phi}{\partial z}(r, \theta, h, t) = 0\ . \tag{4}$$

What a curious problem! First we have a wave governed by the Laplace equation, and then we have the time-dependence entering through a boundary condition.

There is one additional quantity of importance, and this is the deviation of the perturbed free surface from the equilibrium free surface at $z = h$. We denote this height deviation by $\eta(r, \theta, t)$. It may be calculated, once the potential Φ is known, from the formula

$$\eta(r,\,\theta,\,t) = -\frac{1}{g}\frac{\partial\Phi}{\partial t}(r,\,\theta,\,h,\,t) \quad . \tag{5}$$

3. Normal Modes of Oscillation

We look for normal modes of oscillation -- i.e., standing waves with pure sinusoidal time dependence. Thus we try

$$\Phi(r,\,\theta,\,z,\,t) = \cos(\omega t)\,\Psi(r,\,\theta,\,z) \quad . \tag{6}$$

Then Ψ satisfies

$$\frac{1}{r}\frac{\partial}{\partial r}\left(r\frac{\partial\Psi}{\partial r}\right) + \frac{1}{r^2}\frac{\partial^2\Psi}{\partial\theta^2} + \frac{\partial^2\Psi}{\partial z^2} = 0\,,\;\; 0 < r < a,\; 0 \le \theta \le 2\pi,\; \text{and}\; 0 < z < h\,, \tag{7}$$

$$\frac{\partial\Psi}{\partial z}(r,\theta,\,0) = 0,\;\; \text{and}\; \frac{\partial\Psi}{\partial r}(a,\,\theta,\,z) = 0, \tag{8}$$

and, from equation (4),

$$-\omega^2\,\Psi(r,\,\theta,\,h) + g\frac{\partial\Psi}{\partial z}(r,\,\theta,\,h) = 0\;. \tag{9}$$

As we shall see, this last equation determines the frequencies of the modes.

As discussed in class, the problem specified by equations (7) and (8) may be solved by separation of variables. The result (after a straightforward but lengthy analysis) is

$$\Psi_{n,m}(r,\,\theta,\,z) = J_m\left(\alpha_n{}^{(m)}\,r/a\right)\{A_m\cos(m\theta) + B_m\sin(m\theta)\}\cosh\left(\alpha_n{}^{(m)}\,z/a\right) \quad . \tag{10}$$

Here $\alpha_n{}^{(m)}$ is the nth positive root of $J_m{}' = 0$. The functions in equation (10) satisfy equation (7) and the boundary conditions (8). By substituting this solution for Ψ into equation (9), we get an equation determining the frequency of the mode:

$$\omega_{n,m}{}^2 = \frac{g\alpha_n{}^{(m)}}{a}\tanh\left(\alpha_n{}^{(m)}\,h/a\right) \quad . \tag{11}$$

The quantity $\omega_{n,m}$ is the angular frequency in radians per second. The linear frequency is $\nu = \omega/2\pi$, so the linear frequencies in Hz are

$$\nu_{n,m} = \frac{1}{2\pi}\left\{\frac{g\alpha_n{}^{(m)}}{a}\tanh\left(\alpha_n{}^{(m)}\,h/a\right)\right\}^{1/2} \;. \tag{12}$$

4. Modal Frequencies

We are now going to evaluate the frequencies of the lower modes for a particular coffee cup. We will need the zeros of the derivatives of the Bessel functions. This capability was available in *Mathematica* 5, but is not directly available in *Mathematica* 7. *Mathematica* 7 does have a built-in function to give zeros of the Bessel function itself, but not zeros of the derivative. The built-in function is BesselJZero[m,n] which returns the nth positive zero of J_m. We use this to construct a function BesselJPrimeZero[m,n] which returns the nth zero of J_m. The code for this is annotated and tested elsewhere (in the notebook derbesszer.nb), so we just include the code without annotation here.

In[1]:=
```
BesselJPrimeZero[m_, n_] := Module[{left, right, z},
  right = N[BesselJZero[m, n]]; Which[(n > 1), (left = N[BesselJZero[m, n - 1]];
    Re[z /. Flatten[FindRoot[D[BesselJ[m, z], z] == 0, {z, left, right}]]]),
   (n == 1), (If[(m == 0), (0.0), (If[(m < 0.5), (left = √m), (left = 0.5 * right)];
      Re[z /. Flatten[FindRoot[D[BesselJ[m, z], z] == 0, {z, left, right}]]])])]]
```

Now we set some parameter values. My favorite coffee cup has a radius of

In[2]:=
```
a = 0.04; (** cup radius in m **)
```

The height of the coffee in my full cup is

In[3]:=
```
h = 0.08; (** coffee height in m **)
```

The acceleration of gravity is

In[4]:=
```
g = 9.81; (** gravity in m^2/s **)
```

The modal frequency in Hz for a given α is

In[5]:=
```
ν[α_] := (1 / (2 * π)) * Sqrt[(g * α / a) * Tanh[α * h / a]]
```

Now we will calculate the first five frequencies (n = 1, 2, 3, 4, 5) for each of the first five angular modes (m = 0, 1, 2, 3, 4). We begin by arranging the 25 relevant α values in a matrix, with each row being the five roots for a fixed m.

In[6]:=
```
row1 = Table[BesselJPrimeZero[0, i], {i, 1, 5}]
```

Out[6]=
```
{0., 3.83171, 7.01559, 10.1735, 13.3237}
```

In[7]:=
```
row2 = Table[BesselJPrimeZero[1, i], {i, 1, 5}]
```

Out[7]=
```
{1.84118, 5.33144, 8.53632, 11.706, 14.8636}
```

In[8]:=
```
row3 = Table[BesselJPrimeZero[2, i], {i, 1, 5}]
```

Out[8]=
```
{3.05424, 6.70613, 9.96947, 13.1704, 16.3475}
```

In[9]:=
```
row4 = Table[BesselJPrimeZero[3, i], {i, 1, 5}]
```

Out[9]=
```
{4.20119, 8.01524, 11.3459, 14.5858, 17.7887}
```

In[10]:=
```
row5 = Table[BesselJPrimeZero[4, i], {i, 1, 5}]
```

Out[10]=
```
{5.31755, 9.2824, 12.6819, 15.9641, 19.196}
```

The zero value for the first root when $m = 0$ is a real mode. The frequency is zero, so it is not an interesting wave mode, but it corresponds to a constant function in the Fourier-Bessel expansion, quite analogous to the constant term in a cosine series.

Now we form the α matrix of roots.

In[11]:=
```
αmat = {row1, row2, row3, row4, row5}
```

Out[11]=
```
{{0., 3.83171, 7.01559, 10.1735, 13.3237}, {1.84118, 5.33144, 8.53632, 11.706, 14.8636},
 {3.05424, 6.70613, 9.96947, 13.1704, 16.3475},
 {4.20119, 8.01524, 11.3459, 14.5858, 17.7887}, {5.31755, 9.2824, 12.6819, 15.9641, 19.196}}
```

Then to get the nth root of J_m', we type αmat[[m+1,n]]. For example, the second root of J_3' is

In[12]:=
```
αmat[[4, 2]]
```

Out[12]=
```
8.01524
```

Note that we have to shift the m-value by 1, because the m's start at zero and the first row of the matrix is row 1. This is a potential source of error later if we forget to carry out the shift, so we define a function now that lets us use the